AD-A128 332 TEKPLOT: A COMPUTER PROGRAM FOR INTERACTIVE VIEWING OF TIME SERIES DATA A..(U) DEFENCE RESEARCH ESTABLISHMENT PACIFIC VICTORIA (BRITISH COLU.. P M HOLTHAM DEC 81 1/1

UNCLASSIFIED DREP-TM-81-11 F/G 9/2 NL



END
DATE FILMED
6 83
DTIC



Victoria, B.C.

Technical Memorandum 81-11

TEKPLOT
A COMPUTER PROGRAM FOR INTERACTIVE VIEWING OF TIME SERIES DATA AND ITS SPECTRAL CONTENT

P.M. Holtham

December 1981

Research and Development Branch
Department of National Defence
Canada

DEFENCE RESEARCH ESTABLISHMENT PACIFIC
VICTORIA, B.C.

DREP Technical Memorandum 81-11

TEKPLOT

A COMPUTER PROGRAM FOR INTERACTIVE VIEWING OF TIME SERIES DATA AND ITS SPECTRAL CONTENT.

P. M. Holtham

December 1981

Approved by:

Chief/DREP

RESEARCH AND DEVELOPMENT BRANCH
DEPARTMENT OF NATIONAL DEFENCE
CANADA

ABSTRACT

A computer program is described which enables on-line display of chosen sections of time-series data files. Files can be stored either on magnetic tape or on disk, and two files from the same unit can be displayed simultaneously. The program is fully interactive, flexible, and simple to use. Default options are used throughout whenever possible. Fourier transforms can also be computed, and the resultant power spectra plotted. Examples are presented of both the output display and program usage.

1. INTRODUCTION

In experiments where data are recorded either continuously or for long periods of time, it is fairly clear that vast amounts of information can soon be accumulated. Only in a few cases, such as perhaps the study of some ambient property, is analysis of all the available data required. In most cases, only small sections of the data are relevant, but the remainder must be recorded to ensure that the significant events are themselves logged. This is particularly true when studying relatively infrequent short-lived natural phenomena whose times of occurrence are not known in advance.

There is thus a need for a means of readily and speedily isolating the comparatively short data sections containing the 'relevant events' and discriminating against the much longer but less significant background data.

Batch processing is not well suited for such selection, since any one particular portion of data may have to be viewed under several different conditions to decide if it is really of use. Furthermore, in more basic research, there may well be little preconceived idea of the form of the event for which one is searching, thus making the development of automatic scanning algorithms impracticable. On-line operation, on the other hand, is ideal for such previewing and hence for the selection of data segments that merit more detailed analysis. An additional benefit, apart from the obvious saving in time, is that the immediate response gives the user a much better feel for his data, and thus aids in his selection.

The purpose of the present article therefore, is to present a computer program that enables such on-line previewing, and which we have found useful in our own investigations of long term geophysically related data. The user can move at will to different sections of his data, replot interesting sections using different scale conditions, and as a further aid, request a Fourier transform and view the resultant power spectrum. To aid in comparison between different measurements, two time series can be manipulated simultaneously. The program is currently being run using a Tektronix CRT terminal connected to a Sigma 7 computer operating under CP-V. Little

difficulty is foreseen in transferring the program to an alternative system that supports Calcomptype calls for on-line video terminals. Areas that are system dependent have been placed in subroutines and are discussed below.

The following sections discuss the use of the program and describe the routines of TEKPLOT. In the appendix we give flow-charts for the two largest routines and give a listing of the complete program.

2. PROGRAM USAGE

Before discussing the program itself, it is convenient to give an overview of the use of TEKPLOT. As will be seen, the program has been designed for a minimum of user input, and consequently, a number of default values have been set up. Provided the user is satisfied with the individual defaults, he need only enter a carriage return (hereafter CR) in response to several questions. Certain other options can be ignored, and scaling conditions chosen automatically, by simple CR responses. Sample input and plots are shown in Figures 1-3, and a list of the necessary input parameters, together with their default values, is given in Table 1.

2.1. INITIALISATION

2.1.1. FILE SPECIFICATION

After starting the program, the user is prompted for the name of the file from which data are to be read and for the number and label of the tape on which the file resides. Entering 9998 or 9999 at this point† informs the program that the file resides on disk. CR defaults to disk (9998).

The user is then prompted for a second file which is assumed to be on the same unit. If only one file is desired, then CR is entered.

---

† In the present system, the number 9998 refers to a disk used for permanent storage, whereas 9999 indicates a temporary storage disk

---



Figure 1. Screen Display of Typical Input Parameters.

```
 ENTER FILE NAME IN 2A4
?X-01M

 ENTER TAPE , I4, 3A4, CR DEFAULTS TO 9998
?

 FILE 2   CR = NONE
?XTESTT

 ENTER NSTART, TIME LENGTH,  AND SECS PER PAGE
 DEFAULTS ARE 1, 20,  AND    10  FORMAT I10
?30.
?12.
?12.

 ENTER SAMPLE RATE IN I5, CR IS 40 SAMP PER SEC
?45.

 ENTER START TIME OF TAPE   CR TO IGNORE THIS OPTION.   12,213
?12 30 05

 ENTER SCALING    CR FOR PER PAGE, 1 FOR WHOLE DISPLAY, 2 TO BE INPUT
?

 SCALE PER PAGE
 TAPE  9998      FILE X-01M
                      XTESTT
 TAPE START      12 30  5
 START  AT            30     NUMBER OF POINTS      540
 SECONDS PER PAGE     12     DATA POINTS PER INCH  54.00


 ENTER CR TO CONTINUE, A TO ABORT ?
```

Figure 2. An Example of the Time Series Display.



Figure 3. An Example of the Power Spectra Display.

Table 1. Input Parameters and Default Values

Variable	Purpose	Default
IFILE	name of first file	-
NTAPE	tape number and label	9998 (disk)
IFILE2	second file name	option ignored
NSTART	starting point in file	1
NDISP	total number of seconds to display	20
NSEC	seconds to display per page	10
ISP	sample rate per second	40
IHST, IMST, ISST	start time of tape	option ignored
ISCALE	scaling mode	0
additional parameters whenever used		
YS	range per division for time series plots	-
PLTMIN	minimum for power spectra plot	-
PLTSTP	range per division (in dB) for plot of spectra plot	-

2.1.2. DATA SEGMENT SPECIFICATION

The next four parameters refer respectively to the point in the file from which to start plotting, the total number of seconds of data to be plotted, the number of seconds of data to be plotted per page, and the sampling rate of the data in points per second (which must be integral). CR response to any of these prompts results in the displayed default value being substituted.

2.1.3. TIME OPTION

To the next prompt, concerning the time option, the starting time of the data file can be entered. Thereafter, each page of the plot will indicate the time of the start of the particular segment being displayed. If time keeping is not required, simply enter CR.

2.1.4. TIME SERIES SCALING OPTIONS

In response to the next prompt, three alternatives can be given which affect the manner in which the time series is scaled before plotting.

In mode 0 scaling (CR entered) each page is treated independently, and the scaling conditions are chosen automatically so that the plot remains in the range of the screen. Furthermore, if two files are being plotted, each time series is scaled independently. This mode is the most useful when the two time series have markedly different amplitudes, and when the data have a wide dynamic range.

In mode 1 scaling (a response of 1) the maximum and minimum values are found for the whole of the file one data segment defined in 2.1.2. These values are then used to determine the upper and lower edges of every time series plot. The second file is plotted using the same scaling factors. This mode is used when many pages of a time series have to be used with the same scaling, or when a quick visual comparison of two files is required.

In mode 2 scaling the user is prompted to enter the value appropriate to the ordinate division. The program then positions the plot automatically so that the minimum of the plot lies at the bottom of the y-axis. This positioning is done independently for each page, and for each plot on the page if two files are being plotted. (It should be noted that when two time series are being plotted, then the scale factor input is doubled before use.)

2.2. POST PLOT INPUT

After the above parameters are entered, the user is given a pause to review his input. Typing CR instigates the plot, whereas entering 'A' returns the program to 2.1.2 and requests a new time series.

When each page of the time series plot is complete, the user has four choices. Entering CR causes the program to continue with the next page of the time series, whereas the 'A' option causes all remaining plots of the time series to be ignored and the program to return to 2.1.2. The 'F' option causes a Fourier Transform to be computed, and the power spectrum plotted using default scaling factors. Power is plotted in dB, and the frequency range is from d.c. to one half of the sampling frequency.

The 'S' option also plots the power spectrum, but the user is first prompted for the minimum and the range per division of the power axis. These values then become the default values for subsequent 'F' calls and remain unchanged until a further 'S' call.

2.3. POST POWER SPECTRA INPUT

On completion of a power spectrum plot the program pauses. Entry of CR at this point causes the program to move to the next page of the time series. Entry of 'A' returns the program to 2.1.2.

When all pages of the specified part of the time series are exhausted, the program prompts for a further segment (section 2.1.2). If no further plots are required, and end of file (ESC F) will terminate the run.

3. THE FOURIER TRANSFORM

The Fourier Transform routine invoked is a conventional single sided fast transform. Before transforming, the mean is subtracted from the data, and the data tapered by a sine squared function. This tapering is applied over a range of 10% of the data, taken at each end of the data (see Figure 4). Zeroes are appended whenever necessary to increase the number of data points to an integral power of two.



Figure 4. The tapering applied in the POWSPC routine. a) shows the sine squared window, and b) shows a cosine data train, of length 240 elements, after tapering.

4. PROGRAM DESCRIPTION

In this section we present a description of the various routines used in TEKPLOT. Supplementary description can be found in the flow charts and listing of Appendix A. Any areas of subroutines that may contain system dependent code are identified.

4.1. MAIN PROGRAM

Two of the principal arrays, DATA and DAT2, are used to contain data appropriate to one page of the time series for file one and file two respectively. The array X1 is used to contain the generated abscissa values needed for plotting. These arrays are dimensioned NMAX + 2, where NMAX is the largest number of points per page that the program is configured to use. For our application an NMAX of 4096 was satisfactory, but this could be increased provided sufficient core is available.

After opening the plotting routines and clearing the screen, the program prompts the user for pertinent information concerning the time series to be plotted. The responses are subsequently analysed for CR default replies, and the necessary default parameters set up.

Various other parameters and control variables are then set up. These include three logical variables, DOTIME, TWOPLT and DISK, which indicate that the time option is being used, that two files are being analysed simultaneously, and that the original data files were stored on disk.

Parameter set up is followed by relevant parameter printout, and then a pause to enable the user to check the variables. A CR response causes the program to continue, whereas an 'A' response causes the program to return to statement number 884 and request the input parameters once again.

The program then checks to see if two files are being used, and if so, whether they reside on disk. If the files are found to be on magnetic tape, then file two is first copied to disk. This copy prevents multiple rewinding of the tape during the plotting sequence.

The program then checks the scale mode, and if mode one is being used, then the maximum and minimum of the first time series are found, and the appropriate scaling factors determined for later use. During this process, if file one was stored on tape, the time series is copied to disk. This markedly increases efficiency by avoiding duplicate tape input during the plotting loop.

The program then enters the main plotting loop. Provided that the previous page was not the last page of the specified time series, a new page of data is read into the array DATA, and if necessary into DAT2. The abscissa values are computed and stored in the array X1. The scale mode is again checked, and the minimum of the plot and the range per y-division are computed and stored in positions NDO + 1 and NDO + 2 of the two data arrays. Throughout this portion of code, NDO is the number of time series points (for each file) to be plotted on the present page. Headers are then written, axes formed, and the present page plotted.

The user is now asked whether he wishes a Fourier transform performed, whether he simply wishes to continue with the time series plots, or whether he wishes to abort the entire process. As may be seen from the listing, only the responses 'F' and 'S' cause a branch to code that has not already been discussed. In the case of the 'S' response, the user is prompted for two variables, namely PLTMIN and PLTSTP, which represent the minimum value of the power spectra plot (in decibels), and the range per division. After clearing the screen, the power spectra are computed, frequency values are set

up in X1, headers and axes are plotted, and the power spectra displayed. We note that the number of spectral points being plotted is NDO2, which may not equal one half of NDO since the time domain data may have to be padded to meet the fast Fourier transform requirement of the number of points equalling a power of two. The 'F' response is the same as for 'S' except that the user is not prompted for PLTMIN and PLTSTP.

The program then pauses to permit the user to view the spectra, and waits for a response. A response of CR causes the program to go back around the main plotting loop, whereas a response of 'A' causes any remaining plotting to be ignored, and the program to return to statement number 884.

Finally, at statement number 998, the user has finished with the program, the initial input unit is closed, any temporary files created by TEKPLOT are deleted, and the code exited.

4.2. SUBROUTINE POWSPC

This routine is responsible for providing a fast Fourier transform, calculating the power spectrum, and expressing it in decibels.

The routine first finds the mean from all of the data, and then tapers the data with a sine squared function. This function has value zero at the end points of the data, and has increased to unity over a range of 10% of NDO within each end of the data. The routine then finds the lowest number L which is a power of two and greater than or equal to the number of data points. The data of length NDO is then padded with zeroes until it is of length L.

The exact call to the fast Fourier transform routine is system dependent and may have to be altered for different installations. In our particular system, the call includes the area from which data are to be taken, the exponent K where 2**K = L, an array S into which the internal sine table will be placed, a variable indicating that the sine table is to be calculated, and an error flag IFERR.

On return from the transform call, error conditions are checked for, the power spectrum computed, and the result expressed in dB. The routine then exits.

4.3. SUBROUTINE PLTOPEN

This routine may require some additional lines of code in some installations. In its present form, a call to PLTNAME initialises parameters within the plot package, and then clears the screen. The call to FINITT forces the present plot buffer to be output, and leaves the screen in alphanumeric mode at coordinate (0,770). Finally the call to limit informs the plot package of the maximum logical page size.

4.4. SUBROUTINE NWPLOT

This routine has entry points NWPLOT and NUPAGE. Both entries clear the screen and provide a suitable time lapse before returning to the user. NWPLOT first pauses, requesting user input, before clearing the screen. A CR response causes a normal return to the calling routine, whereas an 'A' response causes a return to the statement indicated in the calling argument.

4.5. SUBROUTINES HEADER AND FTAXES

These two routines contain the coding necessary to put most of the headers, numbers and symbols, on both the time series and power spectra plots. Except for AXISM, all subroutine calls are standard CALCOMP calls. AXISM is a common modification of AXIS which suppresses non-significant zeroes from the axis numerals before writing them.

The entry point HEADR2 is provided to enable output of solely the second file name.

4.6. SUBROUTINE CLOSE

This subroutine may not be necessary in some other installations. Its purpose is to close the initial input unit, and delete any temporary files that may have been created by TEKPLOT.

4.7. SUBROUTINES BREAD AND BWRITE

These two routines, which read and write elements of a data file, will almost certainly need some modification to satisfy the protocol for handling files in a particular installation.

The routine BREAD is used to read a specified number of elements, starting at a particular specified point in a file, and transfer them into an array. Within a particular transfer, the elements are sequential, but a subsequent call to BREAD might ask for the transfer of a block from a completely different part of the file. The routine handles reads from both tape and disk. In the present implementation, the routine is structured to read from permanent disk files (ITAPE(1) = 9998), or temporary disk files (ITAPE(1) = 9999).

The routine BWRITE essentially performs the inverse function of BREAD, transferring from an array into an output file. The coding however, need only support write operations to the temporary disk (9999), since the subroutine is never called upon to write to permanent disk.

Any system dependent calls, needed to support tape labels, file formats, file searching etc., should be placed in these two routines.

APPENDIX A

1. Flow Chart for TEKPLOT Main Routine




set up plot scaling factors
plot one page of the time series.
prompt user for required action
A
yes
abort?
no
B
skip the F.F.T.?
yes
no
user to specify spectral plotting parameters?
yes
read plot parameters.
Calculate F.F.T and plot the resultant power spectra.
B
Branch back for a new page.
end of file
User finished with program.
Close units and files appropriately then terminate.

Flow Chart for Power Spectrum Routine



3 PROGRAM LISTING

```
C
C
C     TEKPLOT
C                                          P.M. HOLTHAM
C                                          D.R.E.P.
C                                          VICTORIA, B.C.
C                                          CANADA
C
C                                          NOVEMBER 1981
C
C
C      TEKPLOT ENABLES INTERACTIVE VIEWING OF UP TO TWO
C      FILES CONTAINING TIME SERIES DATA.   FILES CAN RESIDE
C      EITHER ON MAGNETIC TAPE OR ON DISK.
C
C      THE USER CAN VIEW SELECTED PORTIONS OF HIS DATA, AND
C      AND CAN CHOOSE FURTHER SEGMENTS WHICH ARE
C      EITHER FURTHER AHEAD, OR BEHIND, IN THE FILE.   SEVERAL
C      SCALING MODES ARE AVAILABLE FOR PLOTTING THE DATA
C
C      FOURIER TRANSFORMS OF THE SEGMENTS CAN BE CALCULATED,
C      AND THE RESULTANT POWER SPECTRA PLOTTED.
C
C
       REAL DATA(4098),X1(4098),DAT2(4098)
       COMMON NDO2,IDUM,DATA,X1,DAT2
       LOGICAL DOTIME,TWOPLT,DISK
       DIMENSION NTAP2(5),TEMPDAT(4)
       INTEGER NTAPE(5),IFILE(5),IFILE2(5)
       COMMON /HDR/ NTAPE,IFILE,IFILE2,NST,IH,IM,IS,NSEC,DOTIME
       COMMON /FTCOMM/ PLTMIN,PLTSTP,AXS,ISP,TWOPLT
       DATA NTAP2(1)/9999/
       INTEGER A/'A'/,FEFF/'F'/,ISPAC/' '/
       DATA RMIN,RMAX/Z7FFFFFFF,Z80000001/
C
C
C
C     SET UP MAXIMUM NUMBER OF POINTS PER PAGE
       NMAX=4096
C
C
C     OPEN THE PLOTTER AND CLEAR THE SCREEN
       CALL PLTOPEN
C
C
C     PROMPT THE USER FOR INPUT PARAMETERS AND READ THEM
C     PARAMETERS ARE:
C       FIRST FILE NAME
C       TAPE OR DISC NUMBER
C       SECOND FILE NAME
C       START POINT IN THE FILE(S)
```

```
52.     C       TOTAL NUMBER OF SECONDS OF DATA TO BE PLOTTED
53.     C       NUMBER OF SECONDS TO BE PLOTTED ON EACH PAGE
54.     C       SAMPLE RATE OF DATA (INTEGER)
55.     C       STARTING TIME OF THE BEGINNING OF FILE(S)
56.     C       SCALING MODE
57.     C       SCALE PARAMETERS IF SCALE MODE EQUALS 2
58.     C
59.             WRITE(6,191)
60.     191     FORMAT( ' ENTER FILE NAME IN 2A4')
61.             READ(5,83) (IFILE(I),I=1,2)
62.     83      FORMAT(2A4)
63.             WRITE(6,292)
64.     292     FORMAT(/' ENTER TAPE , I4, 3A4, CR DEFAULTS TO 9998')
65.             READ(5,293) (NTAPE(J),J=1,4)
66.     293     FORMAT(I4,3A4)
67.             WRITE(6,202)
68.     202     FORMAT(/' FILE 2.  CR = NONE')
69.             READ(5,83) (IFILE2(I),I=1,2)
70.     884     WRITE(6,811)
71.     811     FORMAT(/' ENTER NSTART, TIME LENGTH,  AND SECS PER PAGE',
72.            1 /' DEFAULTS ARE 1, 20,  AND    10',',  FORMAT I10')
73.             READ (5,80,END=998) NSTART,NDISP,NSEC
74.     80      FORMAT(I10)
75.             WRITE(6,294)
76.     294     FORMAT(/' ENTER SAMPLE RATE IN I5, CR IS 40 SAMP PER',
77.            1 ' SEC')
78.             READ(5,295) ISP
79.     295     FORMAT(I5)
80.             WRITE(6,771)
81.     771     FORMAT(/' ENTER START TIME OF TAPE.  CR TO IGNORE',
82.            1  ' THIS OPTION.   I2,2I3')
83.             READ(5,772) IHST,IMST,ISST
84.     772     FORMAT(I2,2I3)
85.     777     WRITE(6,774)
86.     774     FORMAT(/' ENTER SCALING.   CR FOR PER PAGE,'
87.            1 ' 1 FOR WHOLE DISPLAY, 2 TO BE INPUT')
88.             READ(5,775) ISCALE
89.     775     FORMAT(I1)
90.     C      LOOP BACK IF ILLEGAL SCALE MODE
91.             IF((ISCALE.EQ.0).OR.(ISCALE.EQ.1).OR.(ISCALE.EQ.2))GOTO 877
92.             GOTO 777
93.     877     CONTINUE
94.             IF (ISCALE.EQ.1) WRITE(6,778)
95.     778     FORMAT(/' SCALE THROUGH THE WHOLE DATA ')
96.             IF (ISCALE.EQ.0) WRITE(6,779)
97.     779     FORMAT(/' SCALE PER PAGE ')
98.     C
99.             IF (ISCALE.NE.2) GOTO 878
100.    C      ENTER SCALING PARAMETERS IF ISCALE=2
101.            WRITE(6,879)
102.    879     FORMAT(/' ENTER Y SCALE, UNITS PER INCH, F10.5')
103.            READ(5,890) YS
104.    890     FORMAT(F10.5)
105.            WRITE(6,891) YS
```

```
106.    891   FORMAT(/' Y SCALE WILL BE ',G12.5,' UNITS PER INCH',/
107.         1 ,' Y MIN WILL BE CHOSEN AUTOMATICALLY FOR EACH GRAPH')
108.    878   CONTINUE
109.    C
110.    C
111.    C
112.    C
113.    C     SET UP DEFAULT, CONTROL AND SUNDRY OTHER PARAMETERS
114.          IF (NTAPE(1).EQ.0) NTAPE(1)=9998
115.          DISK=.FALSE.
116.          IF ((NTAPE(1).EQ.9999).OR.(NTAPE(1).EQ.9998)) DISK=.TRUE.
117.          TWOPLT=.TRUE.
118.          IF (IFILE2(1).EQ.ISPAC) TWOPLT=.FALSE.
119.          IF (ISP.EQ.0) ISP=40
120.          DOTIME=.TRUE.
121.          IF (IHST+IMST+ISST.EQ.0) DOTIME=.FALSE.
122.    C     NPNTS IS TOTAL NUMBER OF POINTS TO BE DISPLAYED
123.          NPNTS=NDISP*ISP
124.          IF (NSTART.EQ.0) NSTART=1
125.          IF (NPNTS.EQ.0) NPNTS=800
126.          IF (NSEC.EQ.0) NSEC=10
127.    C     ISTART IS START TIME OF TAPE EXPRESSED IN SECONDS
128.          ISTART=60*(IHST*60+IMST)+ISST
129.    C     PAGE CONTROL PARAMETERS
130.    C     NOIN IS NUMBER OF INCHES PER PAGE
131.    C     NPOINT IS NUMBER OF POINTS TO BE PLOTTED EACH PAGE
132.    C     PERIN IS THE POINT DENSITY PER INCH
133.          NOIN=10
134.          NPOINT=NSEC*ISP
135.    C    CHECK NPOINT NOT TOO LARGE
136.          IF (NPOINT.LE.NMAX) GOTO 100
137.          CALL NUPAGE
138.          WRITE(6,101) NPOINT
139.    101   FORMAT(/ ' NPOINT OF ',I20,' TOO LARGE. RUN ABORTED')
140.          CALL NWPLOT(884S)
141.          GOTO 884
142.    C
143.    100   PERIN=NPOINT*0.1
144.    C     SET UP DEFAULT POWER SPECTRA PLOT PARAMEYERS
145.          PLTMIN=-60.
146.          PLTSTP=10.
147.    C     SET UP AXS EQUAL TO NUMBER OF PLOTS TO BE DONE
148.          AXS=1
149.          IF (TWOPLT) AXS=2
150.    C
151.    C    OUTPUT PARAMETERS FOR USER REFLECTION
152.    C
153.          WRITE(6,84) (NTAPE(I),I=1,1),IFILE(1),IFILE(2)
154.    84    FORMAT( ' TAPE ',I5 ,'     FILE ',2A4)
155.          IF (TWOPLT) WRITE(6,203) (IFILE2(I),I=1,2)
156.    203   FORMAT(21X,2A4)
157.          IF (DOTIME) WRITE (6,773) IHST,IMST,ISST
158.    773   FORMAT( ' TAPE START ',3X,3I3)
159.          WRITE(6,88) NSTART,NPNTS,NSEC,PERIN
```

```
 88    FORMAT( ' START  AT ',5X,I8,'    NUMBER OF POINTS ',I8,
     1   /' SECONDS PER PAGE ',I5,'    DATA POINTS PER INCH ',
     2  F6.2)
C
C
C
C
C
C     IF FILE IS NOT ALREADY ON DISK, COPY FILE2 FROM
C     TAPE TO TEMPORARY DISK, TO AVOID
C     MULTIPLE TAPE REWINDS
C
      J=0
      IF (.NOT.TWOPLT) GOTO 398
      IF (DISK) GOTO 398
      IF (NPNTS.LT.4096) GOTO 361
C     TRANSFER IN BLOCKS OF 4096
      DO 353 I=1,NPNTS,4096
      IF (I+4096-1.GT.NPNTS) GOTO 361
      J=J+1
      CALL BREAD(NTAPE,IFILE2,4096,DAT2,I-1+NSTART)
      CALL BWRITE(NTAP2,IFILE2,4096,DAT2,I-1+NSTART)
353   CONTINUE
361   CONTINUE
      NUM=NPNTS-J*4096
      IF (NUM.EQ.0) GOTO 398
C     TRANSFER ANY REMAINDER
      CALL BREAD(NTAPE,IFILE2,NUM,DAT2,J*4096+NSTART)
      CALL BWRITE(NTAP2,IFILE2,NUM,DAT2,J*4096+NSTART)
398   CONTINUE
C
C
C
C
C     HANDLE SCALE MODE = 1 SITUATION
C     LOOK FOR MAXIMUM AND MINIMUM VALUES OF THE SPECIFIED
C     TIME SERIES FOR FILE ONE
      IF (ISCALE.NE.1) GOTO 801
      J=0
      IF (NPNTS.LT.4096) GOTO 261
C     SCAN THROUGH DATA IN BLOCKS OF 4096
      DO 253 I=1,NPNTS,4096
      IF ( I+4096-1 .GT.NPNTS ) GOTO 261
      J=J+1
      CALL BREAD(NTAPE,IFILE,4096,DATA,I-1+NSTART)
C     COPY FILE ONE TO DISK IF IT WAS NOT ALREADY THERE
      IF (.NOT.DISK)  CALL BWRITE(NTAP2,IFILE,4096,DATA,I-1+NSTART)
      DO 257 K=1,4096
      IF (DATA(K).GT.RMAX) RMAX=DATA(K)
      IF (DATA(K).LT.RMIN) RMIN=DATA(K)
257   DATA(K)=1.E35
253   CONTINUE
261   CONTINUE
      NUM=NPNTS-J*4096
```

```
214.     C    SCAN THROUGH ANY REMAINING POINTS
215.          IF (NUM.EQ.0) GOTO 198
216.          CALL BREAD(NTAPE,IFILE,NUM,DATA,J*4096+NSTART)
217.          IF(.NOT.DISK) CALL BWRITE(NTAP2,IFILE,NUM,DATA,J*4096+NSTART)
218.          DO 259 K=1,NUM
219.          IF (DATA(K).GT.RMAX) RMAX=DATA(K)
220.          IF (DATA(K).LT.RMIN) RMIN=DATA(K)
221.     259  DATA(K)=1.E35
222.     198  CONTINUE
223.     C    CALCULATE THE MINIMUM, AND THE RANGE PER DIVISION
224.     C    NEEDED FOR THE PLOT
225.          TEMPDAT(1)=RMAX
226.          TEMPDAT(2)=RMIN
227.          CALL SCALE (TEMPDAT,8.,2,1)
228.          YMIN=TEMPDAT(3)
229.          YSEP=TEMPDAT(4)
230.          WRITE(6,1974) RMIN,RMAX
231.     1974 FORMAT( ' MIN AND MAX SIGNAL       ',2E15.5)
232.          WRITE(6,150) YMIN,YSEP
233.     150  FORMAT( ' MINIMUM Y AND SEPARATION ',2E15.5)
234.     801  CONTINUE
235.     C
236.     C
237.     C
238.     C
239.     C
240.     C    BEGINNING OF THE DATA PLOTTING LOOP
241.     C    TWO ENTRY POINTS: 10 AND 110, DEPENDING ON WHETHER
242.     C    CLEAR PAGE NEEDS A PAUSE OR NOT
243.          START=NSTART
244.          NDO=NPOINT
245.          TOTDON=0
246.          NST=NSTART-NPOINT
247.          J=0
248.     10   CALL NWPLOT(884S)
249.          GOTO 111
250.     110  CALL NUPAGE
251.     C    INCREASE NUMBER OF POINTS DONE BY A PAGE
252.     111  TOTDON=TOTDON+NPOINT
253.     C    IF LAST PAGE PLOTTED WAS NOT FULL, THEN MUST HAVE FINISHED
254.     C    THE TIME SERIES.  GO BACK AND ASK FOR ANOTHER
255.     C    TIME SERIES SEGMENT
256.          IF (NDO.LT.NPOINT) GOTO 884
257.     C    INCREASE STARTING POINT BY ONE PAGE
258.          NST=NST+NPOINT
259.     C    IF A COMPLETE PAGE WOULD TAKE US PAST THE END OF
260.     C    THE SPECIFIED TIME SERIES, ADJUST NUMBER OF POINTS
261.          IF (TOTDON.GT.NPNTS) NDO=NPOINT-(TOTDON-NPNTS)
262.     C    IF NDO IS ZERO, THEN THE WHOLE TIME SERIES SEGMENT
263.     C    HAS BEEN PLOTTED EXACTLY, WITHOUT THE NEED FOR A
264.     C    SHORT PAGE.   GO BACK AND ASK FOR A NEW TIME SERIES
265.     C    SEGMENT
266.          IF (NDO.EQ.0) GOTO 884
267.     C    INCREASE POINTER TO END OF DATA IN CURRENT PLOT
```

```
268.           NFIN=NST+NDO-1
269.     C
270.     C     READ THE DATA FOR FILE ONE
271.           IF ((.NOT.DISK).AND.(ISCALE.EQ.1))
272.          1 CALL BREAD(NTAP2,IFILE,NDO,DATA,NST,0)
273.           IF ((DISK).OR.((.NOT.DISK).AND.(ISCALE.NE.1)))
274.          1 CALL BREAD(NTAPE,IFILE,NDO,DATA,NST)
275.     C
276.     C     READ DATA FOR SECOND FILE IF NEEDED
277.           IF (.NOT.TWOPLT) GOTO 814
278.           IF(.NOT.DISK) CALL BREAD(NTAP2,IFILE2,NDO,DAT2,NST,0)
279.           IF(DISK) CALL BREAD(NTAPE,IFILE2,NDO,DAT2,NST)
280.     814   CONTINUE
281.     C
282.     C     SET UP X-AXIS POINTS AND THE SCALE INFORMATION
283.           I=0
284.           DO 20 IK=NST,NFIN
285.           I=I+1
286.           X1(I)=IK
287.     20    CONTINUE
288.           X1(NDO+1)=FLOAT(NST)
289.           X1(NDO+2)=PERIN
290.     C
291.     C
292.     C
293.     C     SET UP THE SCALING APPROPRIATE TO THE THREE MODES
294.     C
295.           IF (ISCALE.NE.1) GOTO 804
296.     C     MODE 1
297.     C     ALL PAGES OF BOTH FILES ARE SCALED IDENTICALLY.
298.     C     SCALING IS APPROPRIATE TO THE MAXIMUM AND MINIMUM VALUES
299.     C     OF THE WHOLE OF THE SPECIFIED TIME SERIES OF FILE ONE
300.           DATA(NDO+1)=YMIN
301.           DATA(NDO+2)=YSEP
302.           DAT2(NDO+1)=YMIN
303.           DAT2(NDO+2)=YSEP
304.           GOTO 806
305.     C
306.     C
307.     804   IF (ISCALE.NE.0) GOTO 805
308.     C     MODE 0
309.     C     EVERY PAGE  IS SCALED SEPARATELY AND AUTOMATICALLY.
310.     C     BOTH FILES ARE SCALED INDEPENDENTLY
311.           CALL SCALE(DATA,8.,NDO,1)
312.           IF(TWOPLT) CALL SCALE(DAT2,8.,NDO,1)
313.           GOTO 806
314.     C
315.     C
316.     C     MODE 2
317.     C     RANGE PER DIVISION WAS READ IN, AND IS USED THROUGHOUT
318.     C     MINIMA ARE FOUND SEPERATELY FOR EACH PAGE AND FOR EACH FILE
319.     805   CALL SCALE(DATA,8.,NDO,1)
320.           IF (TWOPLT) CALL SCALE(DAT2,8.,NDO,1)
321.           DATA(NDO+2)=YS
```

```
      DAT2(NDO+2)=YS
C
C
C     HALVE THE SCALE IF TWO PLOTS TO BE DONE
806    IF (TWOPLT) DATA(NDO+2)=DATA(NDO+2)*2.
       IF (TWOPLT) DAT2(NDO+2)=DAT2(NDO+2)*2.
C
C     CALCULATE TIME AT START OF CURRENT PAGE
       NSECDON= (NST-1)/ISP      + ISTART
       IH=NSECDON/3600
       ILEFT=NSECDON-IH*3600
       IM=ILEFT/60
       IS=ILEFT-IM*60
       IF (IH.GE.24) IH=IH-24
C
C     WRITE HEADER FOR THE TIME SERIES PLOT
       CALL HEADER(TWOPLT)
       CALL SYMBOL(7.50,7.0,.15,'SCALE  MODE',.0,11)
       CALL NUMBER(9.5,7.0,.15,FLOAT(ISCALE  ),.0,0)
C
C
C     DO THE TIME SERIES PLOTS
       IF (TWOPLT) CALL PLOT(.0,4.,-3)
C     DRAW THE AXIS FOR THE FIRST FILE DATA PLOT
C     EIGHT INCHES IF ONE PLOT, THREE INCHES IF TWO PLOTS
       AXL=8.
       IF (TWOPLT) AXL=3.
       CALL AXISM(.0,.0,' ',1,AXL,90.,DATA(NDO+1),DATA(NDO+2)
      1 ,.15,.15)
C     PLOT THE DATA
       CALL LINE (X1,DATA,NDO,1,0,0)
       CALL FINITT(0,0)
C
 CC
C     DO THE SECOND FILE IF REQUIRED
       IF (.NOT.TWOPLT) GOTO 11
       CALL PLOT(.0,-4.,-3)
C     DRAW THE AXIS FOR THE SECOND DATA PLOT
       CALL AXISM(.0,.0,' ',1,AXL,90.,DAT2(NDO+1),DAT2(NDO+2)
      1, .15  15)
C     PLOT THE DATA
       CALL LINE(X1,DAT2,NDO,1,0,0)
11      CONTINUE
        CALL FINITT(0,0)
C
C
C     THAT'S THE TIME SERIES PLOTTING FINISHED, WHAT NEXT.
C
C     FOURIER TRANSFORM REQUIRED?
        WRITE(6,12)
12      FORMAT(' ENTER CR FOR CONTINUE, F FOR F.T., S YOU SCALE',
      1 ', A FOR ABORT')
        CALL FINITT(860,0)
        READ(5,13,END=998) IRESP
```

```
376.     13      FORMAT(A1)
377.     C
378.     C     A = IGNORE REMAINING OF PLOTS, CR = IGNORE F.F.T.
379.     C     F OR S IMPLIES DO A TRANSFORM
380.           IF (IRESP.EQ.'A') CALL NUPAGE
381.           IF (IRESP.EQ.'A') GOTO 894
382.           IF (IRESP.EQ.ISPAC) GOTO 110
383.           IF (IRESP.EQ.'FFF') GOTO 1012
384.     C
385.     C     USER MUST SUPPLY POWER SPECTRA PLOT SCALING
386.           CALL NUPAGE
387.           WRITE(6,1013)
388.     1013  FORMAT(' ENTER PLOT MINIMUM IN DB.  F10.5')
389.           READ(5,1014) PLTMIN
390.     1014  FORMAT(F10.5)
391.           WRITE(6,1015)
392.     1015  FORMAT(' ENTER PLOT SCALE PER INCH IN DB.  F10.5')
393.           READ(5,1014) PLTSTP
394.     C
395.     C     CALCULATE THE POWER SPECTRUM AFTER CLEARING THE SCREEN
396.     1012  CALL NUPAGE
397.           CALL POWSPC(DATA,NDO)
398.     C     PUT IN THE SCALING
399.           DATA(NDO2+1)=PLTMIN
400.           DATA(NDO2+2)=PLTSTP*AXS
401.     C
402.     C
403.     C     PUT HEADER AND AXES ON F.T. PLOTS
404.     C     FORCE THE SECOND FILE NAME TO BE OMITTED
405.           CALL HEADER(.FALSE.)
406.           CALL FTAXES
407.     C     PUT ON ANY SECOND FILENAME AFTER PLOT HAS BEEN
408.     C     FACTORED BY FTAXES CALL
409.           IF (TWOPLT) CALL HEADR2
410.     C
411.     C
412.     C     SET UP THE X-AXIS AND SCALING
413.           DO 703 IK=1,NDO2
414.     703   X1(IK)=.5*FLOAT(ISP)*FLOAT(IK-1)/FLOAT(NDO2)
415.           X1(NDO2+1)=0.
416.           X1(NDO2+2)=.05*ISP
417.     C
418.     C     PLOT THE FIRST POWER SPECTRUM
419.           IF (TWOPLT) CALL PLOT(.0,4.5,-3)
420.           CALL LINE(X1,DATA,NDO2,1,0,0)
421.     C
422.     C     CALCULATE AND PLOT THE SECOND FILE SPECTRUM IF NEEDED
423.           IF (.NOT.TWOPLT) GOTO 41
424.           CALL POWSPC(DAT2,NDO)
425.           DAT2(NDO2+1)=PLTMIN
426.           DAT2(NDO2+2)=PLTSTP*AXS
427.           CALL FINITT(30,30)
428.           CALL PLOT(.0,-4.5,-3)
429.           CALL LINE(X1,DAT2,NDO2,1,0,0)
```

```
C
C
C     POWER SPECTRA COMPLETED. GO BACK AROUND THE PAGE LOOP
41     CALL FINITT(30,30)
         GOTO 10
C
C
C     USER HAS FINISHED WITH THE PROGRAM
C      CLOSE INPUT UNIT AND DELETE ANY TEMPORARY FILES
C      CREATED BY TEKPLOT
998    CALL CLOSE(NTAPE,NTAP2,DISK)
C
       STOP
       END
```

```
       SUBROUTINE  POWSPC(A,NDO)
       DIMENSION A(1),S(2056)
       COMMON NDO2
       COMMON /FTCOMM/ PLTMIN,PLTSTP,AXS,ISP,TWOPLT
C
C
C
C      SUBTRACT OUT DC AFTER FINDING THE MEAN
       SUM=0.
       DO 3 IK=1,NDO
       SUM=SUM+A(IK)
3      CONTINUE
       SUM=SUM/NDO
       DO 4 IK=1,NDO
4      A(IK)=A(IK)-SUM
C
C      WINDOW DATA OVER 10% EACH END WITH SIN**2 WINDOW
       NWIND=NDO/10 + 1
C      SKIP IF NO POINTS TO WINDOW
       IF (NWIND.EQ.0) GOTO 5555
       DO 255 IS=1,NWIND
       RSIN=SIN(.5*3.1415927*(IS-1)/(NWIND-1))
       RSIN=RSIN*RSIN
       A(IS)=A(IS)*RSIN
255    A(NDO+1-IS)=A(NDO+1-IS)*RSIN
C
C      FIND POWER OF 2 GREATER OR EQUAL TO NDO
5555   L=1
       DO 1 I=1,20000
       K=I
       L=2*L
       IF (L.GE.NDO) GOTO 2
  1    CONTINUE
C      L IS AN EXACT POWER OF 2, AND IS EITHER EQUAL TO NDO
C      OR THE NEXT POWER OF 2 ABOVE NDO
C
```

```
C       PAD WITH ZEROES IF L WAS NOT EQUAL TO NDO
C       NUMZ=L-NDO
        IF (NUMZ.EQ.0) GOTO 55
        NDO1=NDO+1
        DO 5 IK=NDO1,L
    5   A(IK)=0.
C
C       CALCULATE THE FFT OF ARRAY A, 2**K ELEMENTS EXACTLY
C       S IS STORAGE FOR THE SINE TABLE
C       -1 SAYS CACULATE THE SINE TABLE EACH TIME
C       IFERR IS AN ERROR FLAG.     0 = OK
C                                  -1 = SINE TABLE ALREADY DONE
55      CALL CORRFFT(A,K,S,-1,IFERR)
C
C       CHECK FOR ANY ERROR CONDITION
       IF (.NOT.((IFERR.EQ.0).OR.(IFERR.EQ.-1))) WRITE(6,6) IFERR
6       FORMAT(/' ERROR ',I2,' IN F.T. ***************')
C
C       SET UP NUMBER OF FOURIER COEFFICIENTS TO PLOT
        NDO2= L/2 + 1
C
C       CALCULATE THE POWER, AND BOTTOM JUSTIFY IT IN ARRAY A
        A(1)=A(1)**2
        TEMP=A(2)**2
        DO 7 IK=3,L,2
        A((IK+1)/2)= (A(IK)**2+A(IK+1)**2)
7       CONTINUE
        A(NDO2)=TEMP
C
C
C      FORM POWER IN DB
        DO 8 IK=1,NDO2
        A(IK)=10.*ALOG10(A(IK))
C
8      IF (A(IK).LT.PLTMIN) A(IK)=PLTMIN
        RETURN
        END
```

```
      SUBROUTINE PLTOPEN
C
C  THIS SUBROUTINE INITIALIZES PLOTTING VARIABLES
C  AND CLEARS THE SCREEN
C
      CALL PLTNAME('MYNAME',6)
      CALL FINITT(0,770)
      CALL LIMIT(0.,10.,0.,14.)
      RETURN
      END
```

```
      SUBROUTINE NWPLOT(*)
       INTEGER A/'A'/
      INTEGER ISPAC/' '/
       CALL FINITT(0,0)
       WRITE(6,10)
10     FORMAT(' ENTER CR TO CONTINUE, A TO ABORT')
       CALL FINITT(480,0)
       READ(5,12) ICR
12     FORMAT(A1)
      GOTO 20
      ENTRY NUPAGE(*)
      ICR=ISPAC
20    CONTINUE
      CALL FINITT(0,48)
      CALL NEWPAG
      CALL TSEND
      CALL PLTNAME(0,0)
       CALL LIMIT(0.,10.,.0,14.)
       IF (ICR.EQ.A) RETURN 1
      RETURN
      END
```

```
      SUBROUTINE HEADER(TWOPLT)
      DIMENSION NTAPE(5),IFILE(5),IFILE2(5)
      LOGICAL TWOPLT,DOTIME
      COMMON /HDR/ NTAPE,IFILE,IFILE2,NST,IH,IM,IS,NSEC,DOTIME
      CALL SYMBOL(4.5,7.5,.15, 'FILE ',.0,5)
      CALL SYMBOL(6.,7.5,.15,IFILE(1),.0,8)
      CALL SYMBOL (7.5,7.5,.15,'TAPE ',.0,5)
      CALL NUMBER (8.5,7.5,.15,FLOAT(NTAPE(1)),.0,0)
      CALL SYMBOL(4.5,7.25,.15, 'NSTART ',.0,8)
      CALL NUMBER(6.,7.25,.15,FLOAT(NST),.0,-1)
      IF (.NOT.DOTIME) GOTO 800
      CALL SYMBOL (7.5,7.25,.15,'TIME ',.0,5)
      CALL NUMBER (8.5,7.25,.15,FLOAT(IH),.0,-1)
      CALL NUMBER (9.,7.25,.15,FLOAT(IM),.0,-1)
      CALL NUMBER (9.50,7.25,.15,FLOAT(IS),.0,-1)
800   CONTINUE
      CALL SYMBOL(4.5,7.0,.15,'NSEC ',.0,5)
      CALL NUMBER(6.,7.,.15,FLOAT(NSEC),.0,-1)
C
C    IF ONLY ONE PLOT TO DO THEN RETURN
C    ELSE WRITE THE SECOND FILE NAME
C
       IF (.NOT.TWOPLT) RETURN
C
C    ENTRY POINT FOR JUST THE SECOND FILE NAME
C
       ENTRY HEADR2
       CALL SYMBOL (4.5,3.5,.15,'FILE ',.0,5)
```

```
      CALL SYMBOL (6.,3.5,.15,IFILE2(1),.0,8)
      RETURN
      END
```

```
      SUBROUTINE FTAXES
      LOGICAL TWOPLT
      COMMON /FTCOMM/ PLTMIN,PLTSTP,AXS,ISP,TWOPLT
      CALL FACTOR(0.8)
      CALL PLOT(1.,.5,-3)
      CALL AXISM(0.,0.,'DB',2,8./AXS,90.,PLTMIN,AXS*PLTSTP,.15,.15)
      CALL AXIS(0.,0.,'FREQ',-4,10.,0.,0.,.05*ISP)
C
C     DRAW SECOND AXES IF TWO PLOTS, ELSE RETURN
C
        IF (.NOT.TWOPLT) RETURN
C
        CALL PLOT(.0,4.5,-3)
      CALL AXISM(.0,.0,'DB',2,8./AXS,90.,PLTMIN,AXS*PLTSTP,.15,.15)
        CALL AXIS(.0,.0,'FREQ',-4,10.,0.,0.,.05*ISP)
        CALL PLOT(.0,-4.5,-3)
      RETURN
      END
```

```
      SUBROUTINE CLOSE(NTAPE,NTAP2,DISK)
C
C     THIS SUBROUTINE IS SYSTEM DEPENDENT AND IS NEEDED
C     IF ANY SPECIAL ACTION HAS TO BE TAKEN TO CLOSE THE
C     INPUT UNIT OR THE TEMPORARY DISK IF USED.
C
C     NTAPE IS THE INPUT UNIT, NTAP2 IS THE TEMPORARY DISK
C     AND THE LOGICAL VARIABLE DISK IS 'TRUE' IF TEMPORARY
C     DISK WAS USED.
C
C     THE FOLLOWING CODE SHOULD BE CHANGED AS REQUIRED
C
      LOGICAL DISK
      DIMENSION NTAPE(5),NTAP2(5)
C
C     CLOSE AND CLEAN TEMPORARY DISK IF USED
      IF (.NOT.DISK) CALL CLOSE1(0,NTAP2)
C
C     CLOSE THE INPUT UNIT
      CALL CLOSE1(1,NTAPE)
C
      RETURN
      END
```

```
1.            SUBROUTINE BREAD(ITAPE,IFIL,NUM,AREA,NSTART)
2.      C
3.      C     THIS SUBROUTINE READS NUM ELEMENTS FROM FILE 'IFIL'
4.      C     STARTING AT ELEMENT NSTART.  THE ELEMENTS ARE TRANSFERRED
5.      C     INTO THE REAL ARRAY 'AREA'
6.      C
7.      C     THE FILE RESIDES ON TAPE OR DISK AS SPECIFIED BY ITAPE(1)
8.      C     INT THE PRESENT SYSTEM, THE MEDIA IS AS FOLLOWS
9.      C
10.     C       ITAPE(1) = 9998        PERMANENT DISK FILE
11.     C       ITAPE(1) = 9999        FILE ON TEMPORARY DISK
12.     C       OTHER VALUES           FILE ON TAPE
13.     C
14.     C     IN THE CASE OF A TAPE, THE TAPE LABEL IS STORED IN
15.     C     ITAPE(2-3). IN ALL CASES THE FILENAME IS STORED IN IFIL(1-2)
16.     C
17.     C     NOTE. CALLS TO 9999 CAN ARISE WHEN READING A TEMPORARY FILE
18.     C     CREATED BY TEKPLOT, OR IF THE USERS ORIGINAL DATA FILE
19.     C     RESIDED ON THE TEMPORARY DISK
20.     C
21.     C     THIS SUBROUTINE IS SYSTEM DEPENDENT, AND THE FOLLOWING CODE
22.     C     SHOULD BE REPLACED BY THE NECESSARY DATA READING CALLS
23.     C
24.           DIMENSION ITAPE(5),IFIL(5),AREA(1)
25.     C     WORDS ITAPE(5) AND IFIL(3-5) USED ONLY IN THIS SYSTEM
26.           COMMON /M1READ/ IENTER
27.           IENTER=1
28.           ITAPE(5)=0
29.     C
30.     C     SEPARATE 9999 FROM OTHER CASES
31.           IF (ITAPE(1).EQ.9999) GOTO 1
32.           CALL DMREAD(1,ITAPE,IFIL,NUM,AREA,NSTART,0)
33.           RETURN
34.     C
35.     C     HANDLE 9999 READS
36.     1       CALL DMREAD(0,ITAPE,IFIL,NUM,AREA,NSTART,0)
37.           RETURN
38.           END
```

```
1.            SUBROUTINE BWRITE(ITAPE,IFIL,NUM,AREA,NSTART)
2.      C
3.      C     THIS SUBROUTINE IS THE INVERSE OF BREAD, AND WRITES NUM
4.      C     ELEMENTS FROM A REAL ARRAY 'AREA' ONTO A FILE 'IFIL'
5.      C     STARTING AT POSITION 'NSTART' IN THE FILE.
6.      C
7.      C     AS IN BREAD, ITAPE(1) WOULD SPECIFIY WHERE THE FILE RESIDES,
8.      C     BUT IN TEKPLOT, BWRITE IS ALWAYS CALLED WITH ITAPE(1)=9999,
9.      C     IE. FILES ARE ONLY WRITTEN ONTO TEMPORARY DISK.
10.     C
```

```
11.     C     THIS SUBROUTINE IS ALSO SYSTEM DEPENDENT, AND THE FOLLOWING
12.     C     CODE SHOULD BE CHANGED TO THAT APPROPRIATE
13.     C
14.           DIMENSION ITAPE(5),IFIL(5),AREA(1)
15.           COMMON /M1READ/ IENTER
16.           IENTER=1
17.           ITAPE(5)=0
18.     C
19.     C     DO THE TRANSFER
20.           CALL DMRITE(0,ITAPE,IFIL,NUM,AREA,NSTART,0)
21.           RETURN
22.           END
```

REPORT NO: DREP TECHNICAL MEMORANDUM 81-11

TITLE: TEKPLOT A COMPUTER PROGRAM FOR INTERACTIVE VIEWING OF TIME SERIES DATA AND ITS SPECTRAL CONTENT

DATED: DECEMBER 1981

AUTHOR: P.M. HOLTHAM

SECURITY GRADING UNCLASSIFIED INITIAL DISTRIBUTION DECEMBER 1981

3- DSIS
Plus distribution
1- DSIS Report Collection
1- Document Section (microfiche)
1- DREV
1- DREP
1- DREO
1- DCIEM
1- DREA

1- CRAD/DST(SE)
1- CRAD/DTA(A)
1- CRAD/DTA(M)
1- DGIS/DSTI
1- DMES

1- ORAE LIBRARY
1- DEEM
1- DMCS
1- DLR
1- DAR
1- DGMEM
1- DNBCC
1- DMRS
1- AETE, COLD LAKE
1- CFANS
1- DMA

1- MARCOM (MAG HQ)
1- MARCOM (MC/ORD)
1- MH PAC (SSO OP RES)
2- NRC LIBRARY
1- LAND TECH. LIBRARY

1- NDHQ TECH. LIBRARY
1- MARITIME TECH. LIBRARY
1- DMEE
1- DCEM
1- DGMPO
1- DEW
3- CSE
1- CFMWS
1- RMC
1- RRMC

BRITAIN
MINISTRY OF DEFENCE
3- DRIC

UNITED STATES
3- DTIC

Unclassified

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall document is classified)

1 ORIGINATING ACTIVITY Defence Research Establishment Pacific FMO, Victoria, B.C. V0S 1B0	2a. DOCUMENT SECURITY CLASSIFICATION Unclassified 2b. GROUP

3 DOCUMENT TITLE

TEKPLOT A COMPUTER PROGRAM FOR INTERACTIVE VIEWING OF TIME SERIES DATA AND ITS SPECTRAL CONTENT

4 DESCRIPTIVE NOTES (Type of report and inclusive dates)

Technical Memorandum

5. AUTHOR(S) (Last name, first name, middle initial)

HOLTHAM, P.M.

6 DOCUMENT DATE December 1981	7a. TOTAL NO. OF PAGES 32	7b. NO. OF REFS
8a. PROJECT OR GRANT NO. 37A	9a. ORIGINATOR'S DOCUMENT NUMBER(S) 81-11	
8b CONTRACT NO.	9b. OTHER DOCUMENT NO.(S) (Any other numbers that may be assigned this document)	

10. DISTRIBUTION STATEMENT

Qualified requesters may obtain copies of this document from their defence documentation center.

11 SUPPLEMENTARY NOTES	12. SPONSORING ACTIVITY

13. ABSTRACT

A computer program is described which enables on-line display of chosen sections of time-series data files. Files can be stored either on magnetic tape or on disk, and two files from the same unit can be displayed simultaneously. The program is fully interactive, flexible, and simple to use. Default options are used throughout whenever possible. Fourier transforms can also be computed, and the resultant power spectra plotted. Examples are presented of both the output display and program usage.

DSIS
78-070

KEY WORDS

Computer Program
Interactive Viewing
Time Series Plotting
Time Series Analysis
Spectral Content

INSTRUCTIONS

1. ORIGINATING ACTIVITY: Enter the name and address of the organization issuing the document.

2a. DOCUMENT SECURITY CLASSIFICATION: Enter the overall security classification of the document including special warning terms whenever applicable.

2b. GROUP: Enter security reclassification group number. The three groups are defined in Appendix 'M' of the DRB Security Regulations.

3. DOCUMENT TITLE: Enter the complete document title in all capital letters. Titles in all cases should be unclassified. If a sufficiently descriptive title cannot be selected without classification, show title classification with the usual one-capital-letter abbreviation in parentheses immediately following the title.

4. DESCRIPTIVE NOTES: Enter the category of document, e.g. technical report, technical note or technical letter. If appropriate, enter the type of document, e.g. interim, progress, summary, annual or final. Give the inclusive dates when a specific reporting period is covered.

5. AUTHOR(S): Enter the name(s) of author(s) as shown on or in the document. Enter last name, first name, middle initial. If military, show rank. The name of the principal author is an absolute minimum requirement.

6. DOCUMENT DATE: Enter the date (month, year) of Establishment approval for publication of the document.

7a. TOTAL NUMBER OF PAGES: The total page count should follow normal pagination procedures, i.e., enter the number of pages containing information.

7b. NUMBER OF REFERENCES: Enter the total number of references cited in the document.

8a. PROJECT OR GRANT NUMBER: If appropriate, enter the applicable research and development project or grant number under which the document was written.

8b. CONTRACT NUMBER: If appropriate, enter the applicable number under which the document was written.

9a. ORIGINATOR'S DOCUMENT NUMBER(S): Enter the official document number by which the document will be identified and controlled by the originating activity. This number must be unique to this document.

9b. OTHER DOCUMENT NUMBER(S): If the document has been assigned any other document numbers (either by the originator or by the sponsor), also enter this number(s).

10. DISTRIBUTION STATEMENT: Enter any limitations on further dissemination of the document, other than those imposed by security classification, using standard statements such as:

(1) "Qualified requesters may obtain copies of this document from their defence documentation center."

(2) "Announcement and dissemination of this document is not authorized without prior approval from originating activity."

11. SUPPLEMENTARY NOTES: Use for additional explanatory notes.

12. SPONSORING ACTIVITY: Enter the name of the departmental project office or laboratory sponsoring the research and development. Include address.

13. ABSTRACT: Enter an abstract giving a brief and factual summary of the document, even though it may also appear elsewhere in the body of the document itself. It is highly desirable that the abstract of classified documents be unclassified. Each paragraph of the abstract shall end with an indication of the security classification of the information in the paragraph (unless the document itself is unclassified) represented as (TS), (S), (C), (R), or (U).

The length of the abstract should be limited to 20 single-spaced standard typewritten lines; 7½ inches long.

14. KEY WORDS: Key words are technically meaningful terms or short phrases that characterize a document and could be helpful in cataloging the document. Key words should be selected so that no security classification is required. Identifiers, such as equipment model designation, trade name, military project code name, geographic location, may be used as key words but will be followed by an indication of technical context.